# $Id$
# Maintainer: Jan Alexander Steffens (heftig) <jan.steffens@gmail.com>

pkgname=bijiben
pkgver=3.24.1
pkgrel=1
pkgdesc="Write out notes, every detail matters"
url="https://wiki.gnome.org/Apps/Bijiben"
arch=(i686 x86_64)
license=(GPL)
depends=(tracker zeitgeist webkit2gtk gnome-online-accounts evolution-data-server)
makedepends=(intltool itstool evolution python gnome-common appstream-glib git)
groups=(gnome-extra)
_commit=82b1903e379dcc417b4661acda807b9d156abcb4  # tags/BIJIBEN_3_24_1^0
source=("git+https://git.gnome.org/browse/bijiben#commit=$_commit"
        "git+https://git.gnome.org/browse/libgd")
sha256sums=('SKIP'
            'SKIP')

pkgver() {
  cd $pkgname
  git describe --tags | sed 's/^BIJIBEN_//;s/_/./g;s/-/+/g'
}

prepare() {
  cd $pkgname

  git submodule init
  git config --local submodule.libgd.url "$srcdir/libgd"
  git submodule update

  NOCONFIGURE=1 ./autogen.sh
}

build() {
  cd $pkgname
  ./configure \
    --prefix=/usr \
    --sysconfdir=/etc \
    --localstatedir=/var \
    --libexecdir=/usr/lib/$pkgname \
    --disable-update-mimedb \
    --enable-compile-warnings
  make
}

package() {
  cd $pkgname
  make DESTDIR="$pkgdir" install
}
